content: -- Copyright (c) 2016-present, Facebook, Inc.
  -- All rights reserved.
  --
  -- This source code is licensed under the BSD-style license found in the
  -- LICENSE file in the root directory of this source tree.
  
  
  {-# LANGUAGE GADTs #-}
  {-# LANGUAGE LambdaCase #-}
  {-# LANGUAGE OverloadedStrings #-}
  
  module Duckling.AmountOfMoney.KA.Rules
    ( rules
    ) where
  
  import Data.Maybe
  import Data.String
  import Prelude
  import qualified Data.Text as Text
  
  import Duckling.AmountOfMoney.Helpers
  import Duckling.AmountOfMoney.Types (Currency(..), AmountOfMoneyData (..))
  import Duckling.Dimensions.Types
  import Duckling.Numeral.Helpers (isNatural, isPositive)
  import Duckling.Numeral.Types (NumeralData (..))
  import Duckling.Regex.Types
  import Duckling.Types
  import qualified Duckling.AmountOfMoney.Types as TAmountOfMoney
  import qualified Duckling.Numeral.Types as TNumeral
  
  ruleUnitAmount :: Rule
  ruleUnitAmount = Rule
    { name = "<unit> <amount>"
    , pattern =
      [ Predicate isCurrencyOnly
      , Predicate isPositive
      ]
    , prod = \case
        (Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.currency = c}:
         Token Numeral NumeralData{TNumeral.value = v}:
         _) -> Just . Token AmountOfMoney . withValue v $ currencyOnly c
        _ -> Nothing
    }
  
  rulePounds :: Rule
  rulePounds = Rule
    { name = "£"
    , pattern =
      [ regex "(გირვანქა )?სტერლინგი?ს?(ად)?თ?|(გირვანქა )?ფუნ?ტი?ს?(ად)?თ?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly Pound
    }
  
  ruleOtherPounds :: Rule
  ruleOtherPounds = Rule
    { name = "other pounds"
    , pattern =
      [ regex "(ეგვიპტური|ლიბანური) ?ფუნ?ტი?ს?(ად)?თ?"
      ]
    , prod = \tokens -> case tokens of
        (Token RegexMatch (GroupMatch (match:_)):_) -> case Text.toLower match of
          "ეგვიპტური" -> Just . Token AmountOfMoney $ currencyOnly EGP
          "ლიბანური" -> Just . Token AmountOfMoney $ currencyOnly LBP
          _          -> Nothing
        _ -> Nothing
    }
  
  ruleRiyals :: Rule
  ruleRiyals = Rule
    { name = "riyals"
    , pattern =
      [ regex "(კატარული|საუდის არაბული) ?რიალი?"
      ]
    , prod = \tokens -> case tokens of
        (Token RegexMatch (GroupMatch (match:_)):_) -> case Text.toLower match of
          "კატარული" -> Just . Token AmountOfMoney $ currencyOnly QAR
          "საუდის არაბული"  -> Just . Token AmountOfMoney $ currencyOnly SAR
          _        -> Nothing
        _ -> Nothing
    }
  
  ruleDinars :: Rule
  ruleDinars = Rule
    { name = "dinars"
    , pattern =
      [ regex "(ქუვეითური)? ?დინარი?"
      ]
    , prod = \tokens -> case tokens of
        (Token RegexMatch (GroupMatch (match:_)):_) -> case Text.toLower match of
          "ქუვეითური" -> Just . Token AmountOfMoney $ currencyOnly KWD
          _         -> Nothing
        _ -> Nothing
    }
  
  ruleDirham :: Rule
  ruleDirham = Rule
    { name = "დირჰემი"
    , pattern =
      [ regex "დირჰემი?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly AED
    }
  
  ruleRinggit :: Rule
  ruleRinggit = Rule
    { name = "ringgit"
    , pattern =
      [ regex "(მალაიზი?ური? )?რინგიტი?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly MYR
    }
  
  ruleCent :: Rule
  ruleCent = Rule
    { name = "cent"
    , pattern =
      [ regex "ცენტი?ს?(ად)?თ?|თეთრი?ს?(ად)?თ?|პენსი?ს?(ად)?თ?|ევროცენტი?ს?(ად)?თ?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly Cent
    }
  
  ruleLari :: Rule
  ruleLari = Rule
    { name = "Lari"
    , pattern =
      [ regex "(ლარი?ს?(ად)?თ?)|GEL"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly GEL
    }
  
  ruleDollars :: Rule
  ruleDollars = Rule
    { name = "Dollars"
    , pattern =
      [ regex "დოლარი?ს?(ად)?თ?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly Dollar
    }
  
  ruleEuros :: Rule
  ruleEuros = Rule
    { name = "Euros"
    , pattern =
      [ regex "ევრო?ს?დ?თ?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly EUR
    }
  
  ruleACurrency :: Rule
  ruleACurrency = Rule
    { name = "a <currency>"
    , pattern =
      [ Predicate isCurrencyOnly
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         _) -> Just . Token AmountOfMoney $ fd {TAmountOfMoney.value = Just 1}
        _ -> Nothing
    }
  
  ruleIntersectAndXCents :: Rule
  ruleIntersectAndXCents = Rule
    { name = "intersect (and X cents)"
    , pattern =
      [ Predicate isWithoutCents
      , regex "და"
      , Predicate isCents
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         _:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  ruleIntersectAndNumeral :: Rule
  ruleIntersectAndNumeral = Rule
    { name = "intersect (and number)"
    , pattern =
      [ Predicate isWithoutCents
      , regex "და"
      , Predicate isNatural
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         _:
         Token Numeral NumeralData{TNumeral.value = c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  ruleIntersectXCents :: Rule
  ruleIntersectXCents = Rule
    { name = "intersect (X cents)"
    , pattern =
      [ Predicate isWithoutCents
      , Predicate isCents
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  rulePrecision :: Rule
  rulePrecision = Rule
    { name = "about|exactly <amount-of-money>"
    , pattern =
      [ regex "ზუსტად|იმენა|დაახლოებით|გძეტა"
      , Predicate isMoneyWithValue
      ]
    , prod = \tokens -> case tokens of
        (_:token:_) -> Just token
        _ -> Nothing
    }
  
  ruleIntervalBetweenNumeral :: Rule
  ruleIntervalBetweenNumeral = Rule
    { name = "between|from <numeral> to|and <amount-of-money>"
    , pattern =
      [ Predicate isPositive
      , regex "-დან|დან"
      , Predicate isSimpleAmountOfMoney
      , regex "-ა?მდე|ა?მდე"
      ]
    , prod = \tokens -> case tokens of
        (Token Numeral NumeralData{TNumeral.value = from}:
         _:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just to,
                    TAmountOfMoney.currency = c}:
         _) | from < to ->
          Just . Token AmountOfMoney . withInterval (from, to) $ currencyOnly c
        _ -> Nothing
    }
  
  ruleIntervalBetween :: Rule
  ruleIntervalBetween = Rule
    { name = "between|from <amount-of-money> to|and <amount-of-money>"
    , pattern =
      [ Predicate isSimpleAmountOfMoney
      , regex "-დან|დან"
      , Predicate isSimpleAmountOfMoney
      , regex "-ა?მდე|ა?მდე"
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just from,
                    TAmountOfMoney.currency = c1}:
         _:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just to,
                    TAmountOfMoney.currency = c2}:
         _) | from < to && c1 == c2 ->
          Just . Token AmountOfMoney . withInterval (from, to) $ currencyOnly c1
        _ -> Nothing
    }
  
  ruleIntervalNumeralDash :: Rule
  ruleIntervalNumeralDash = Rule
    { name = "<numeral> - <amount-of-money>"
    , pattern =
      [ Predicate isNatural
      , regex "-"
      , Predicate isSimpleAmountOfMoney
      ]
    , prod = \tokens -> case tokens of
        (Token Numeral NumeralData{TNumeral.value = from}:
         _:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just to,
                    TAmountOfMoney.currency = c}:
         _) | from < to->
           Just . Token AmountOfMoney . withInterval (from, to) $ currencyOnly c
        _ -> Nothing
    }
  
  ruleIntervalDash :: Rule
  ruleIntervalDash = Rule
    { name = "<amount-of-money> - <amount-of-money>"
    , pattern =
      [ Predicate isSimpleAmountOfMoney
      , regex "-"
      , Predicate isSimpleAmountOfMoney
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just from,
                    TAmountOfMoney.currency = c1}:
         _:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just to,
                    TAmountOfMoney.currency = c2}:
         _) | from < to && c1 == c2 ->
          Just . Token AmountOfMoney . withInterval (from, to) $ currencyOnly c1
        _ -> Nothing
    }
  
  ruleIntervalMax :: Rule
  ruleIntervalMax = Rule
    { name = "under/less/lower/no more than <amount-of-money>"
    , pattern =
      [ Predicate isSimpleAmountOfMoney
      , regex "-ა?მდე|ა?მდე|(-ზე|ზე) ნაკლები"
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just to,
                    TAmountOfMoney.currency = c}:
         _) -> Just . Token AmountOfMoney . withMax to $ currencyOnly c
        _ -> Nothing
    }
  
  ruleIntervalMin :: Rule
  ruleIntervalMin = Rule
    { name = "over/above/at least/more than <amount-of-money>"
    , pattern =
      [ Predicate isSimpleAmountOfMoney
      , regex "-დან|დან|(-ზე|ზე) მეტი"
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just to,
                    TAmountOfMoney.currency = c}:
         _) -> Just . Token AmountOfMoney . withMin to $ currencyOnly c
        _ -> Nothing
    }
  
  rules :: [Rule]
  rules =
    [ ruleUnitAmount
    , ruleACurrency
    , ruleDollars
    , ruleCent
    , ruleEuros
    , ruleDinars
    , ruleDirham
    , ruleIntersectAndNumeral
    , ruleIntersectAndXCents
    , ruleIntersectXCents
    , ruleIntervalBetweenNumeral
    , ruleIntervalBetween
    , ruleIntervalMax
    , ruleIntervalMin
    , ruleIntervalNumeralDash
    , ruleIntervalDash
    , ruleOtherPounds
    , rulePounds
    , rulePrecision
    , ruleRinggit
    , ruleRiyals
    , ruleLari
    ]
  